

```

10 ' Programme PC.BAS * AC Soft 1988 *
20 '
30 TRR%=&H3F8:LSR%=&H3FD:MSR%=&H3FE ' Registres du 8250
40 SHELL "MODE COM1:1200,E,8,1 > NUL" ' Init. interface série
50 KEY OFF
60 '
70 ON ERROR GOTO 6000
80 '
90 DIM TAB%(4096) ' Réserve du Buffer
100 '
110 CLS:PRINT "* AMSTRAD CPC --> IBM PC *":PRINT
120 LINE INPUT "Nom du fichier ASCII à recevoir : ",NOM$
130 OPEN NOM$ FOR OUTPUT AS #1
140 PRINT:PRINT "Tapez ESCAPE pour débiter la transmission"
150 K$=INKEY$:IF K$="" THEN 150 ELSE IF ASC(K$)<>27 THEN 150
160 B%=0:PRINT
170 '
180 B%=B%+1:PRINT "Block n°":B%;
190 N%=0:T%=17:GOSUB 4000 ' envoi de XON
195 GOSUB 5000:IF R%<>2 THEN 195 ' Attente STX
200 GOSUB 5000
210 IF R%=3 THEN GOSUB 1000:GOTO 180 ' fin d'un block
220 IF R%=26 THEN 250 ' fin du fichier
230 N%=N%+1:TAB%(N%)=R%:GOTO 200 ' caractère dans le buffer
240 '
250 GOSUB 1000:CLOSE #1 ' fermeture du fichier
260 PRINT:PRINT "Transmission terminée"
270 GOTO 6010 ' Fin du programme
280 '
1000 FOR I%=1 TO N%
1010 PRINT #1,CHR$(TAB%(I%));
1020 NEXT
1030 PRINT "reçu"
1040 RETURN
1050 '
4000 IF (INP(LSR%) AND 32)=0 THEN 4000 ELSE OUT TRR%,T% ' Emission
4010 RETURN
4020 '
5000 IF (INP(LSR%) AND 1)=0 THEN 5000 ELSE R%=INP(TRR%) ' Réception
5010 RETURN
5020 '
6000 BEEP:PRINT "ERREUR:"
6010 PRINT "Autre fichier ? [O/N]"
6020 K$=INKEY$:IF K$="" THEN 6020
6030 IF (K$="O" OR K$="o") THEN RUN ELSE ON ERROR GOTO 0:END
6040 '
7000 ' *** Fin du listing ***

```

Figure 3

```

10 ' CPC,BAS * AC Soft 1998 *
20 '
30 DI:OUT &FAF1,&70:OUT &FAF1,&7F:OUT &FAF1,&7:EI
40 DI:OUT &FAF1,&70:OUT &FAF1,&7F:OUT &FAF1,&7:EI
50 '
60 ON ERROR GOTO 6000
70 '
80 MODE 2:PRINT "* AMSTRAD CPC --> IBM PC *":PRINT
90 LINE INPUT "Nom du fichier ASCII a emettre : ",nom$
100 OPENIN nom$
110 PRINT:PRINT "Attente emission":PRINT
120 GOSUB 1020
130 b%=1:PRINT "Block n.":b%;
140 '
150 WHILE NOT EOF
160 LINE INPUT #9,l$
170 FOR l%=1 TO LEN(l$)
180 t%=ASC(MID$(l$,l%,1)):IF t%<32 THEN t%=32
190 GOSUB 4000:NEXT
200 t%=13:GOSUB 4000:t%=10:GOSUB 4000
210 IF n%>2048 THEN n%=0:GOSUB 1000:PRINT "Block n.":b%; ' fin d'un block
220 WEND
230 '
240 t%=26:GOSUB 4000:CLOSEIN ' fin du fichier
250 PRINT "emis":PRINT:PRINT "Transmission terminee"
260 GOTO 6010
270 '
1000 t%=3:GOSUB 4000 ' Envoi ETX
1010 b%=b%+1:PRINT "emis"
1020 GOSUB 5000:IF r%<>17 THEN 1020 ' Attente XON
1030 t%=2:GOSUB 4000 ' Envoi STX
1040 RETURN
1050 '
4000 IF (INP(&FAF1) AND 1)<>1 THEN 4000 ELSE OUT &FAF0,t%:n%=n%+1
4010 RETURN
4020 '
5000 IF (INP(&FAF1) AND 2)<>2 THEN 5000 ELSE r%=INP(&FAF0)
5010 RETURN
5020 '
6000 PRINT CHR$(7);"ERREUR : "
6010 PRINT "Autre fichier ? [O/N]"
6020 k$=INKEY$:IF k$="" THEN 6020
6030 IF UPPER$(k$)="O" THEN RUN ELSE ON ERROR GOTO 0:END
6040 '
7000 ' *** Fin du listing ***

```

Figure 2

Fichier créé sur GPC (caractères à obtenir sur PC) :

à â é è é ë í î ó ô û ç

Résultat sur PC après transfert (caractères obtenus) :

í ñ √ ² æ á û ò õ ò ñ ç

Lignes à écrire :

```
          (í)          (à)
IF C%=140 THEN C%=133:RETURN
          (ñ)          (ä)
IF C%=164 THEN C%=131:RETURN
```



```

10 ' ACCENTS.BAS * AC Soft 1988 *
20 '
30 ON ERROR GOTO 6000
40 '
50 CLS:PRINT "* Traduction fichiers textes CPC *":PRINT
60 LINE INPUT "Nom du fichier à traiter : ",ANC$
70 LINE INPUT "Nom du fichier résultant : ",NOV$
80 IF NOV$=ANC$ THEN PRINT "Noms différents S.V.P":GOTO 70
90 '
100 OPEN ANC$ FOR INPUT AS #1
110 OPEN NOV$ FOR OUTPUT AS #2
120 '
130 PRINT:PRINT "Traitement en cours ..."
140 WHILE NOT EOF(1)
150 LN$=""
160 LINE INPUT #1,LA$
170 FOR I%=1 TO LEN(LA$)
180 C%=ASC(MID$(LA$,I%,1))
190 GOSUB 1000
200 LN$=LN$+CHR$(C%)
210 NEXT
220 PRINT #2,LN$
230 WEND
240 '
250 CLOSE #1
260 CLOSE #2
270 PRINT "Terminé":GOTO 6010
280 '
1000 IF C%<128 THEN RETURN
1010 IF C%=251 THEN C%=130:RETURN ' √ é
1020 IF C%=164 THEN C%=131:RETURN ' ñ â
1030 IF C%=140 THEN C%=133:RETURN ' î à
1040 IF C%=168 THEN C%=135:RETURN ' ï ç
1050 IF C%=145 THEN C%=136:RETURN ' æ ê
1060 IF C%=160 THEN C%=137:RETURN ' á ë
1070 IF C%=253 THEN C%=138:RETURN ' ð è
1080 IF C%=149 THEN C%=139:RETURN ' ò ï
1090 IF C%=150 THEN C%=140:RETURN ' û î
1100 IF C%=167 THEN C%=147:RETURN ' œ ô
1110 IF C%=169 THEN C%=150:RETURN ' ù
1120 IF C%=252 THEN C%=151:RETURN ' ñ ù
1130 RETURN
1140 '
6000 BEEP:PRINT "ERREUR:"
6010 PRINT "Autre fichier ? [O/N]"
6020 K$=INKEY$:IF K$="" THEN 6020
6030 IF (K$="O" OR K$="o") THEN RUN ELSE ON ERROR GOTO 0:END
6040 '
7000 ' *** Fin du listing ***

```

Figure 5

```

10 ' Programme EMUL.BAS * AC Soft 1988 *
20 '
30 TRR%=&H3F8:LSR%=&H3FD:MSR%=&H3FE ' Registres du 8250
40 OPEN "COM1:1200,E,7,1" AS #1:CLOSE #1 ' Init. interface série
50 DIM TB$(10,2)
60 FOR I%=1 TO 10:READ TB$(I%,1),TB$(I%,2):KEY I%,"":NEXT
70 DATA " Sommaire --> Home ",46
80 DATA " Guide --> Ins ",44
90 DATA " Annulation --> Ct-BS ",45
100 DATA " Correction --> BS ",47
110 DATA " Retour --> PgUp ",42
120 DATA " Suite --> PgDn ",48
130 DATA " Répétition --> End ",43
140 DATA " Envoi --> Enter ",41
150 DATA " Connexion/Fin --> Esc ",49
160 DATA " Exit --> F10 ",00
170 '
180 CLS:KEY OFF
190 LOCATE 1,20:PRINT "
200 LOCATE 2,20:PRINT " Emulateur Minitel * AC Soft 1988 * "
210 LOCATE 3,20:PRINT "
220 FOR I%=1 TO 10:LOCATE I%*2+3,27:PRINT TB$(I%,1):NEXT:OLDI%=1
230 '
240 T$=INKEY$:IF T$="" THEN 240
250 I%=0:IF LEN(T$)=2 THEN 320 ELSE T%=ASC(T$)
260 IF T%=127 THEN I%=3 ' Ct-BS
270 IF T%=8 THEN I%=4 ' BS
280 IF T%=13 THEN I%=8 ' Enter
290 IF T%=27 THEN I%=9 ' Esc
300 IF I%=0 THEN GOSUB 3200 ELSE GOSUB 1000
310 GOTO 240
320 T%=ASC(RIGHT$(T$,1)) ' 2ème Code
330 IF T%=71 THEN I%=1 ' Home
340 IF T%=82 THEN I%=2 ' Ins
350 IF T%=73 THEN I%=5 ' PgUp
360 IF T%=81 THEN I%=6 ' PgDn
370 IF T%=79 THEN I%=7 ' End
380 IF T%=68 THEN COLOR 7,0:CLS:END ' F10
390 IF I%>0 THEN GOSUB 1000
400 GOTO 240
410 '
1000 LOCATE OLDI%*2+3,27:COLOR 7,0:PRINT TB$(OLDI%,1)
1010 LOCATE I%*2+3,27:COLOR 0,7:PRINT TB$(I%,1):OLDI%=I%
1020 T%=19:GOSUB 4000:T%=VAL("&h"+TB$(I%,2)):GOSUB 4000
1030 RETURN
1040 '
3200 IF T%>=32 AND T%<=126 THEN 4000 ' Filtrage des caractères
3210 IF T%=130 THEN T%=66:GOSUB 3340:T%=101:GOTO 4000 ' é
3220 IF T%=131 THEN T%=67:GOSUB 3340:T%= 97:GOTO 4000 ' à
3230 IF T%=133 THEN T%=65:GOSUB 3340:T%= 97:GOTO 4000 ' à
3240 IF T%=135 THEN T%=75:GOSUB 3340:T%= 99:GOTO 4000 ' ç
3250 IF T%=136 THEN T%=67:GOSUB 3340:T%=101:GOTO 4000 ' é
3260 IF T%=137 THEN T%=72:GOSUB 3340:T%=101:GOTO 4000 ' ë
3270 IF T%=138 THEN T%=65:GOSUB 3340:T%=101:GOTO 4000 ' è
3280 IF T%=139 THEN T%=72:GOSUB 3340:T%=105:GOTO 4000 ' i
3290 IF T%=140 THEN T%=67:GOSUB 3340:T%=105:GOTO 4000 ' i
3300 IF T%=147 THEN T%=67:GOSUB 3340:T%=111:GOTO 4000 ' ô
3310 IF T%=150 THEN T%=67:GOSUB 3340:T%=117:GOTO 4000 ' û
3320 IF T%=151 THEN T%=65:GOSUB 3340:T%=117:GOTO 4000 ' û
3330 T%=32:GOTO 4000
3340 IF (INP(LSR%) AND 32)=0 THEN 3340 ELSE OUT TRR%,25 ' Préfixe accents
3350 '
4000 IF (INP(LSR%) AND 32)=0 THEN 4000 ELSE OUT TRR%,T% ' Emission
4010 RETURN
4020 '
5000 ' *** Fin du listing ***

```

Figure 6